

Figure 1: Looking at 21 Sample Points

Interactive Optimizer

This **Processing** program is designed to help find the minimum or maximum of a function,

$$y = f(x),$$

on an interval $[a, b]$ and find the value of x that produces this minimum or maximum value. Figure 1 shows the basic idea underlying this program. In this figure the interval $[a, b] = [-5, 5]$. If you run this program you should see this figure. The program computes the values of a function $f(x)$ for 21 equally spaced sample points in the interval. The minimum appears to be the point in the center of the interval. If you look in the console Figure 2 you will see that the lowest y value for these 21 sample points is 10.198039 at the point $x = 0.0$ and the highest y value for these 21 sample points is 12.450275 at the point $x = -5.0$. The sample points are spaced 0.5 units apart. Based on this figure it looks as if the minimum (which might not be one of the sample points) is in the interval $[-0.5, 0.5]$. If you press the mouse button on the minimum point the program will do the same thing on the interval $[-0.5, 0.5]$ around that point. This will let you narrow down the true minimum. You can repeat this several times narrowing down the location of the




Figure 2: The Program and Console




Figure 3: Example Problem

minimum. The calculations involved all involve some roundoff error. So usually you will never get the exact minimum and after narrowing it down several times the function values will look very strange. The same procedure can be used to find a maximum instead of a minimum.

In these notes we illustrate the use of this program by finding the path followed by a light ray traveling from a target to a curved mirror and then to an eye. See Figure 3. The curved mirror is part of the circle of radius $R = 10$ centered at the origin. This circle can be described by the equation:

$$x^2 + y^2 = R^2.$$

If x is the x -coordinate of a point on the mirror then

$$y = \sqrt{R^2 - x^2}$$

is its y -coordinate. For this problem we are looking at light rays that bounce off the top half of the circle.

The eye is located at the point $(-1, 5)$ and the target is located at the point $(1, 5)$. The point at which the light ray bounces off the mirror is determined by Fermat's Principle, which says that in this situation the light ray chooses the shortest possible path. Drawing a rough sketch we can see that the point at which the light ray bounces off the mirror will be in the interval $[-5, 5]$.

Lines 1-10 of the program specify the radius of the mirror, the locations of the eye and the target and the initial interval $[a, b]$. Lines 15-18 define a function that describes the shape of the mirror. Its input is a the x -coordinate of a point on the mirror and its output is the y -coordinate of the point. This function is used in lines 20-25 that specify the function to be minimized or maximized. These lines are the ones you will modify for other optimization problems.

The program is listed below but you should look at it and run it using **Processing**.

Program Listing

```
1 float radius = 10;           // Radius of circular mirror.
2
3 float eyeX = -1.0;           // The x-coordinate of the eye
4 float eyeY = 5;              // The y-coordinate of the eye
5
6 float targetX = 1.0;         // The x-coordinate of the target
7 float targetY = 5;           // The y-coordinate of the target
8
9 float a = -5;                // The left end of the interval
10 float b = 5;                 // The right end of the interval
11
12 float yLow;                  // The lowest value of y for the 21 sample points
13 float yHigh;                 // The highest value of y for the 21 sample points
14
15 float mirror(float x)        // y-coordinate on mirror for a point
16 {
17     return sqrt(radius * radius - x * x);
18 }
19
20 float fcn(float x)
21 {
22     float eyeToMirror = sqrt((eyeX - x) * (eyeX - x) + (eyeY - mirror(x)) * (eyeY - mirror(x)));
23     float targetToMirror = sqrt((targetX - x) * (targetX - x) + (targetY - mirror(x)) * (targetY - mirror(x)));
24     return eyeToMirror + targetToMirror;
25 }
26
27 void settings()              // This routine runs once, before the setup routine
28 {
29     size(821, 421);
30 }
31
32 void setup()                 // The setup routine is run once when the program (sketch )starts
33 {
34     plot21();                // Plot 21 points in the interval
35 }
36
37 void draw()                  // The draw routine is not used but it is necessary
38 {
39 }
40
41 void plot21()
42 {
43     float x, y, xLowest, xHighest;           // Used for computations
44     background(255);                         // Clear display area to all white
45     stroke(128, 128, 255);                   // Grid lines are green
46     for(int i = 0; i <= 20; i = i + 1)       // Draw vertical grid lines
47     {
48         line(10 + i * 40, 10, 10 + i * 40, 410);
```

```

49  }
50  for(int i = 0; i <= 10; i = i + 1)          // Draw horizontal grid lines
51  {
52      line(10, 10 + i * 40, 810, 10 + i * 40);
53  }
54  yLow = fcn(a);
55  xLowest = a;
56  yHigh = fcn(a);
57  xHighest = a;
58  for(int i = 1; i <= 20; i = i + 1)          // Find the minimum and maximum values of y for the 21
59  {
60      x = a + i * 0.05 * (b - a);            // x-value of this sample point
61      y = fcn(x);                             // y-value of this sample point
62      if(y < yLow)                             // Check if this sample point is below the previous 1
63      {
64          yLow = y;
65          xLowest = x;
66      }
67      if(y > yHigh)                             // Check if this sample point is above the previous h
68      {
69          yHigh = y;
70          xHighest = x;
71      }
72  }
73  print("The lowest y value for the 21 sample points is ");
74  print(yLow);
75  print(" at x = ");
76  println(xLowest);
77  print("The highest y value for the 21 sample points is ");
78  print(yHigh);
79  print(" at x = ");
80  println(xHighest);
81  println("");
82  stroke(0);                                  // The dots at the samplepoints are black
83  fill(0);
84  for(int i = 0; i <= 20; i = i + 1)          // Plot sample points
85  {
86      x = a + i * 0.05 * (b - a);            // x-value of this sample point
87      y = fcn(x);                             // y-value of this sample point
88      circle(xOf(x), yOf(y), 8);
89  }
90 }
91
92 void mousePressed()
93 {
94     float dx = (b - a)/20.0;                // x-interval width
95     int x = round((mouseX - 10.0)/40);
96     a = a + (x - 1) * dx;
97     b = a + 2 * dx;
98     plot21();

```

```

99 }
100
101 int xOf(float x)
102 {
103     return round(10 + 800 * (x - a)/(b - a));           // Return the display area x-coordinate of a p
104 }
105
106 int yOf(float y)
107 {
108     return round(410 - 400 * (y - yLow)/(yHigh - yLow)); // Return the display area y-coordinate of a p
109 }

```